Substitutions diverses

© Pierre Lantagne *(Avril 2018)*
Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2003. L'objectif principal de cette feuille Maple est de rendre l'élève apte à transposer en Maple la technique d'intégration par substitution trigonométrique. Cette transposition sera réalisée avec certaines macro-commandes de la sous-bibliothèque Calculus1 de la bibliothèque Student.

Les exemples développés dans ce document illustreront la manière d'obtenir l'intégrale indéfinie de deux façons:

- une première façon consistera à obtenir directement l'intégrale indéfinie avec la macro-commande int de la bibliothèque principale, question d'observer le mécanisme de la simplification automatique de Maple.
- la seconde façon, quant à elle, consistera à faire un développement pas.à pas. Un tel développement sera réalisé selon les règles d'intégration Rule de *Student*[*Calculus1*].

Pour compléter tout développement pas à pas, la dernière étape consistera à faire une vérification de la réponse finale par dérivation.

Attention:

- S'assurer que la rubrique « Afficher les étiquettes des équations » soit cochée: voir *menu Outils* → *Options*... → *Affichage.*
- S'assurer que le préfixe de numérotation d'étiquettes soit initialisée à 1 avec un schéma de numérotation séquentielle: voir *menu Format* → *Étiquettes d'équation* (*L*) → *Affichage des étiquettes*... (*L*).



- L'insertion d'une étiquette est obtenu avec le raccourci « ctrl + l » (l pour *label* en anglais).

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
  with(Student[Calculus1],Rule,Integrand);
  infolevel[Student[Calculus1]]:=1:
```

$$[Rule, Integrand] \tag{1}$$

Assurance que les étiquettes des équations soient affichées.

```
> interface(labelling=true)
```

$$true \tag{2}$$

Réglage de l'affichage pour les variables conditionnées.

```
> with(Typesetting):
  interface(typesetting=standard); #niveau de composition Maple
  Standard
  interface(showassumed=2);         # Variables avec suppositiosnOhrase
```

$$extended, [extended]$$

$$2 \tag{3}$$

Substitutions diverses

Exemple 1

Calculons l'intégrale indéfinie $\int \frac{1}{\sqrt{3+\sqrt{x}}}\,\mathrm{d}x$.

Avec Maple directement:

```
> f:=x->1/sqrt(3+sqrt(x));
```

$$f := x \to \frac{1}{\sqrt{3+\sqrt{x}}} \tag{4}$$

Posons l'intégrale indéfinie et donnons-lui le nom Problème.

```
> Problème:=Int(f(x),x);
```

$$Problème := \int \frac{1}{\sqrt{3+\sqrt{x}}}\,\mathrm{d}x \tag{5}$$

Obtenons directement cette intégrale indéfinie avec la macro-commande `value`.

```
> Rép_Maple:=Problème=value(Problème)+C;
```

$$Rép\_Maple := \int \frac{1}{\sqrt{3+\sqrt{x}}}\,\mathrm{d}x = \frac{4}{3}\left(3+\sqrt{x}\right)^{3/2} - 12\sqrt{3+\sqrt{x}} + C \tag{6}$$

Maintenant, avec un développement pas à pas jusqu'aux formules de base.

```
> Problème;
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,\mathrm{d}x \tag{7}$$

Traitons l'expression sous le premier radical au dénominateur comme une expression de la forme $a^2 + u^2$, avec $a^2 = 3$ et $u^2 = \sqrt{x}$. Intégrons alors par substitution trigonométrique en posant $\sqrt{x} = 3\tan(\theta)^2$ où $\theta \in \left]-\frac{\pi}{2}, \frac{\pi}{2}\right[$.

```
> Rule[change,sqrt(x)=3*tan(theta)^2,theta]((7));
Creating problem #1

Applying substitution x = 9*tan(theta)^4, theta = arctan(1/3*3^(1/2)*x^
(1/4)) with dx = 36*tan(theta)^3*(1+tan(theta)^2)*dtheta, dtheta = 1/12*
3^(1/2)/x^(3/4)/(1/3*x^(1/2)+1)*dx
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = \int \left(\frac{36\tan(\theta)^5}{\sqrt{3+3\tan(\theta)^2}} + \frac{36\tan(\theta)^3}{\sqrt{3+3\tan(\theta)^2}} \right) d\theta \qquad \textbf{(8)}$$

Simplifions en faisant tenir compte par Maple de l'identité suivante $1 + (\tan^2)(\theta) = (\sec^2)(\theta)$.

```
> Rule[rewrite,1+tan(theta)^2=(sec^2)(theta)]((8));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = \int \frac{12\sqrt{3}\,(\sec(\theta)^2-1)\sec(\theta)^2\tan(\theta)}{\sqrt{\sec(\theta)^2}}\,d\theta \qquad \textbf{(9)}$$

Puisque $\theta \in \left]-\frac{\pi}{2}, \frac{\pi}{2}\right[$, simplifions P3 en conditionnant θ avec $\sec(\theta) > 0$.

```
> simplify((9)) assuming (theta>-Pi/2, theta<Pi/2) ;
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(\int \frac{\sin(\theta)^3}{\cos(\theta)^4}\,d\theta\right) \qquad \textbf{(10)}$$

Intégrons **(10)** avec un changement de variables en posant $u = \cos(\theta)$.

```
> Rule[change,u=cos(theta),u]((10));
Applying substitution theta = arccos(u), u = cos(theta) with dtheta =
-1/(-u^2+1)^(1/2)*du, du = -sin(theta)*dtheta
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(\int \frac{u^2-1}{u^4}\,du\right) \qquad \textbf{(11)}$$

Appliquons les propriétés de l'intégrale indéfinie.

```
> expand((11));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(\int \left(\frac{1}{u^2} - \frac{1}{u^4}\right) du\right) \qquad \textbf{(12)}$$

```
> Rule[sum]((12));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(\int \frac{1}{u^2}\,du + \int \left(-\frac{1}{u^4}\right) du\right) \qquad \textbf{(13)}$$

```
> Rule[`c*`]((13));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(\int \frac{1}{u^2}\,du - \left(\int \frac{1}{u^4}\,du\right)\right) \qquad \textbf{(14)}$$

```
> Rule[power]((14));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(\int \frac{1}{u^2}\,du + \frac{1}{3u^3}\right) \quad \textbf{(15)}$$

```
> Rule[power]((15));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 12\sqrt{3}\left(-\frac{1}{u} + \frac{1}{3u^3}\right) \quad \textbf{(16)}$$

Ayant posé $u = \cos(\theta)$, effectuons cette substitution à rebours.

```
> Rule[revert]((16));
Reverting substitution using u = cos(theta)
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = -4\sqrt{3}\left(\frac{3}{\cos(\theta)} - \frac{1}{\cos(\theta)^3}\right) \quad \textbf{(17)}$$

Ayant posé $\sqrt{x} = 3\tan(\theta)^2$, on a que $\theta = \arctan\left(\sqrt{\frac{\sqrt{x}}{3}}\right)$. Effectuons cette seconde substitution à rebours.

```
> Rule[revert]((17));
Reverting substitution using theta = arctan(1/3*3^(1/2)*x^(1/4))
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = \frac{4}{9}\sqrt{3}\sqrt{3\sqrt{x}+9}\left(\sqrt{x}-6\right) \quad \textbf{(18)}$$

Pour formuler la réponse finale, mettons en évidence $\sqrt{3+\sqrt{x}}$.

```
> combine((18),radical);
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = 3\left(\frac{4}{9}\sqrt{x} - \frac{8}{3}\right)\sqrt{3+\sqrt{x}} \quad \textbf{(19)}$$

```
> normal((19));
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = \frac{4}{3}\left(\sqrt{x}-6\right)\sqrt{3+\sqrt{x}} \quad \textbf{(20)}$$

```
> Rép_finale:=(20)+(0=C);
```

$$Rép\_finale := \int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx = \frac{4}{3}\left(\sqrt{x}-6\right)\sqrt{3+\sqrt{x}} + C \quad \textbf{(21)}$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

(22)

$$\frac{\partial}{\partial x}\left(\frac{4}{3}\left(\sqrt{x}-6\right)\sqrt{3+\sqrt{x}}+C\right)=\frac{2}{3}\frac{\sqrt{3+\sqrt{x}}}{\sqrt{x}}+\frac{1}{3}\frac{\sqrt{x}-6}{\sqrt{3+\sqrt{x}}\sqrt{x}} \quad \textbf{(22)}$$

```
> ``=radnormal(rhs((22)));
```

$$=\frac{1}{\sqrt{3+\sqrt{x}}} \quad \textbf{(23)}$$

La simplification donne bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{1}{\sqrt{3+\sqrt{x}}} \quad \textbf{(24)}$$

Remarque: Le résultat donné automatiquement par Maple un tantinet surprenant. Tout de même, il est possible, bien sûr, de simplifier ce résultat.

```
> Rép_Maple+(0=-C);
```

$$\int \frac{1}{\sqrt{3+\sqrt{x}}}\,dx=\frac{4}{3}\left(3+\sqrt{x}\right)^{3/2}-12\sqrt{3+\sqrt{x}} \quad \textbf{(25)}$$

```
> ``=radnormal(rhs((25)));
```

$$=\frac{4}{3}\sqrt{3+\sqrt{x}}\sqrt{x}-8\sqrt{3+\sqrt{x}} \quad \textbf{(26)}$$

```
> simplify((26));
```

$$=\frac{4}{3}\left(\sqrt{x}-6\right)\sqrt{3+\sqrt{x}} \quad \textbf{(27)}$$

Exemple 2

Calculons l'intégrale indéfinie $\int \frac{2+\sin(x)}{1+\cos(x)}\,dx$.

Avec Maple directement:

```
> f:=x->(2+sin(x))/(1+cos(x));
```

$$f:=x\rightarrow\frac{2+\sin(x)}{1+\cos(x)} \quad \textbf{(28)}$$

Posons l'intégrale indéfinie et donnons-lui le nom Problème.

```
> Problème:=Int(f(x),x);
```

$$\mathit{Problème}:=\int \frac{2+\sin(x)}{1+\cos(x)}\,dx \quad \textbf{(29)}$$

Obtenons directement cette intégrale indéfinie avec la macro-commande *value*.

```
> Rép_Maple:=Problème=value(Problème)+C;
```

$$\mathit{Rép\_Maple}:=\int \frac{2+\sin(x)}{1+\cos(x)}\,dx=2\tan\left(\frac{1}{2}x\right)+\ln\left(\tan\left(\frac{1}{2}x\right)^2+1\right)+C \quad \textbf{(30)}$$

Maintenant, avec un développement pas à pas jusqu'aux formules de base.

```
> Problème;
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx \tag{31}$$

Intégrons avec un changement de variables en posant $x = 2\arctan(u)$.

```
> Rule[change,x = 2*arctan(u),u]((31));
Creating problem #2

Applying substitution x = 2*arctan(u), u = tan(1/2*x) with dx = 2/
(u^2+1)*du, du = (1/2*tan(1/2*x)^2+1/2)*dx
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = \int \left(2+\frac{2u}{u^2+1}\right)du \tag{32}$$

Appliquons les règles de l'intégrale indéfinie.

```
> Rule[sum]((32));
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = \int 2\,du + \int \frac{2u}{u^2+1}\,du \tag{33}$$

```
> Rule[constant]((33));
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2u + \int \frac{2u}{u^2+1}\,du \tag{34}$$

```
> Rule[revert]((34));
Reverting substitution using u = tan(1/2*x)
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}x\right) + \int \frac{2u}{u^2+1}\,du \tag{35}$$

```
> Rule[change,t=u^2+1]((35));
Applying substitution u = (t-1)^(1/2), t = u^2+1 with du = 1/2/(t-1)^
(1/2)*dt, dt = 2*u*du
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}x\right) + \int \frac{1}{t}\,dt \tag{36}$$

```
> Rule[power]((36));
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}x\right) + \ln(t) \tag{37}$$

```
> Rule[revert]((37));
Reverting substitution using t = u^2+1
```

(38)

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}\,x\right) + \ln\left(u^2+1\right) \tag{38}$$

```
> Rule[revert]((38));
Reverting substitution using u = tan(1/2*x)
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}\,x\right) + \ln\left(\frac{2}{1+\cos(x)}\right) \tag{39}$$

```
> simplify((39), ln) assuming real;
```

$$\int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}\,x\right) + \ln(2) - \ln(1+\cos(x)) \tag{40}$$

```
> Rép_finale:=(40)+(0=-ln(2)+C);
```

$$\mathit{Rép\_finale} := \int \frac{2+\sin(x)}{1+\cos(x)}\,dx = 2\tan\left(\frac{1}{2}\,x\right) - \ln(1+\cos(x)) + C \tag{41}$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(2\tan\left(\frac{1}{2}\,x\right) - \ln(1+\cos(x)) + C\right) = \tan\left(\frac{1}{2}\,x\right)^2 + 1 + \frac{\sin(x)}{1+\cos(x)} \tag{42}$$

```
> ``=convert(rhs((42)),sincos);
```

$$= \frac{(1-\cos(x))^2}{\sin(x)^2} + 1 + \frac{\sin(x)}{1+\cos(x)} \tag{43}$$

```
> ``=simplify(rhs((43)));
```

$$= \frac{2+\sin(x)}{1+\cos(x)} \tag{44}$$

La simplification donne bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{2+\sin(x)}{1+\cos(x)} \tag{45}$$

Exemple 3

Calculons l'intégrale indéfinie $\int \frac{1}{1+\sin(x)}\,dx$.

Avec Maple directement:

```
> f:=x->1/(1+sin(x));
```

$$f := x \rightarrow \frac{1}{1+\sin(x)} \tag{46}$$

Posons l'intégrale indéfinie et donnons-lui le nom Problème.

```
> Problème:=Int(f(x),x);
```

(47)

$$Problème := \int \frac{1}{1+\sin(x)}\,dx \quad (47)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande `value`.

```
> Rép_Maple:=Problème=value(Problème)+C;
```

$$Rép\_Maple := \int \frac{1}{1+\sin(x)}\,dx = -\frac{2}{\tan\left(\frac{1}{2}x\right)+1} + C \quad (48)$$

Maintenant, avec un développement pas à pas jusqu'aux formules de base.

```
> Problème;
```

$$\int \frac{1}{1+\sin(x)}\,dx \quad (49)$$

Intégrons avec un changement de variables en posant $x = 2\arctan\left(\frac{v}{2}\right)$.

```
> Rule[change, x=2*arctan(v/2),v]((49));
Creating problem #3

Applying substitution x = 2*arctan(1/2*v), v = 2*tan(1/2*x) with dx = 1/
(1/4*v^2+1)*dv, dv = (tan(1/2*x)^2+1)*dx
```

$$\int \frac{1}{1+\sin(x)}\,dx = \int \frac{4}{v^2+4v+4}\,dv \quad (50)$$

```
> factor((50));
```

$$\int \frac{1}{1+\sin(x)}\,dx = \int \frac{4}{(v+2)^2}\,dv \quad (51)$$

```
> Rule[change,t=v+2,t]((51));
Applying substitution v = -2+t with dv=dt
```

$$\int \frac{1}{1+\sin(x)}\,dx = \int \frac{4}{t^2}\,dt \quad (52)$$

```
> Rule[`c*`]((52));
```

$$\int \frac{1}{1+\sin(x)}\,dx = 4\left(\int \frac{1}{t^2}\,dt\right) \quad (53)$$

```
> Rule[power]((53));
```

$$\int \frac{1}{1+\sin(x)}\,dx = -\frac{4}{t} \quad (54)$$

```
> Rule[revert]((54));
Reverting substitution using t = v+2
```

$$\int \frac{1}{1+\sin(x)}\,dx = -\frac{4}{v+2} \quad \mathbf{(55)}$$

```
> Rule[revert]((55));
Reverting substitution using v = 2*tan(1/2*x)
```

$$\int \frac{1}{1+\sin(x)}\,dx = -\frac{4}{2\tan\left(\frac{1}{2}x\right)+2} \quad \mathbf{(56)}$$

```
> simplify((56));
```

$$\int \frac{1}{1+\sin(x)}\,dx = -\frac{2}{\tan\left(\frac{1}{2}x\right)+1} \quad \mathbf{(57)}$$

```
> Rép_finale:=(57)+(0=C);
```

$$\mathit{Rép\_finale} := \int \frac{1}{1+\sin(x)}\,dx = -\frac{2}{\tan\left(\frac{1}{2}x\right)+1}+C \quad \mathbf{(58)}$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(-\frac{2}{\tan\left(\frac{1}{2}x\right)+1}+C\right) = \frac{2\left(\frac{1}{2}\tan\left(\frac{1}{2}x\right)^2+\frac{1}{2}\right)}{\left(\tan\left(\frac{1}{2}x\right)+1\right)^2} \quad \mathbf{(59)}$$

```
> ``=convert(rhs((59)),sincos);
```

$$= \frac{2\left(\frac{1}{2}\,\frac{(1-\cos(x))^2}{\sin(x)^2}+\frac{1}{2}\right)}{\left(\frac{1-\cos(x)}{\sin(x)}+1\right)^2} \quad \mathbf{(60)}$$

```
> ``=simplify(rhs((60)));
```

$$= \frac{1}{1+\sin(x)} \quad \mathbf{(61)}$$

La simplification donne bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{1}{1+\sin(x)} \quad \mathbf{(62)}$$